

Appendix i

Script for the `get_pprime` programme, written primarily by D. S. Brown (private communication). This should be run in `solarsoft idl`, and access to the SPICE package in the STEREO `solarsoft` tree should be enabled. A detailed explanation of the processes can be found in Chapter 4, Section 4.5.

```
function get_pprime, index, ra, dec

; distance from sc to sun
rr = sqrt(index.haex_obs^2 + index.haey_obs^2 + index.haez_obs^2)

; position of the Earth – get_stereo_coord is a SPICE routine
epos = get_stereo_coord(index.date_avg,'Earth',system='hae')
ex = 1000.*epos(0)/rr
ey = 1000.*epos(1)/rr
ez = 1000.*epos(2)/rr

; position of stereo
scx = index.haex_obs/rr
scy = index.haey_obs/rr
scz = index.haez_obs/rr

; IPS line of sight – convert_stereo_lonlat is a SPICE routine
los = [ra, dec]

convert_stereo_lonlat,index.date_avg,los,'gei','hae',/degrees

vx = sin((90.0-los(1))/!rdeg)*cos(los(0)/!rdeg)
vy = sin((90.0-los(1))/!rdeg)*sin(los(0)/!rdeg)
vz = cos((90.0-los(1))/!rdeg)
```

```
; calculate lambda to minimize |E + lambda*v|^2
lambda = -(vx*ex + vy*ey + vz*ez)/(vx^2 + vy^2 + vz^2)
```

```
; hence calculate p point in heliocentric hae
```

```
px = ex + lambda*vx
```

```
py = ey + lambda*vy
```

```
pz = ez + lambda*vz
```

```
; define u, the vector between s/c and p
```

```
ux = px - scx
```

```
uy = py - scy
```

```
uz = pz - scz
```

```
; calculate mu to minimize |sc + mu*u|^2
```

```
mu = -(ux*scx + uy*scy + uz*scz)/(ux^2 + uy^2 + uz^2)
```

```
; hence calculate p' point in heliocentric hae
```

```
ppx = [scx + mu*ux, 1.0]
```

```
ppy = [scy + mu*uy, 0.0]
```

```
ppz = [scz + mu*uz, 0.0]
```

```
; convert p' point to HI fov
```

```
vv = helio2fov(ppx, ppy, ppz, index)
```

```
; do a test plot to see if sensible
```

```
plot,[0,1],[0,1],xrange=[-1.1,1.1],yrange=[-1.1,1.1],xstyle=1,ystyle=1,/nodata,/isotropic
```

```
; plot Sun, Earth, s/c
```

```
oplot,[0.0,ex,scx],[0.,ey,scy],psym=2,symsize=1.5
```

```
; plot p-point, p' point
oplot,[px,ppx(0)],[py,ppy(0)],psym=1,symsize=1.5
```

```
; plot Thomson Sphere for stereo
tsr = sqrt(scx^2 + scy^2 + scz^2)/2.0
th = findgen(181)*!dpi/90.0
cth = 0.5*scx + tsr*cos(th)
sth = 0.5*scy + tsr*sin(th)
oplot,cth,sth,linestyle=1
```

```
; plot Thomson Sphere for IPS
itsr = sqrt(ex^2 + ey^2 + ez^2)/2.0
cth = 0.5*ex + itsr*cos(th)
sth = 0.5*ey + itsr*sin(th)
oplot,cth,sth,linestyle=2
```

```
; plot s/c - p point line
oplot,[scx,px],[scy,py],linestyle=3
```

```
aa = where((vv(0,*) ge 0.0) and (vv(0,*) le 1024) and (vv(1,*) ge 0.0) and
(vv(1,*)le 1024), naa)
```

```
if (naa eq 0) then begin
    out = [-999.9, -999.9]
endif else begin
    out = vv(*,aa)
endelse
```

```
out=vv(*,0)
return,out
end
```